Steps to Success with Data Analysis & Reporting Using PROC MEANS

Andrew H. Karp
Sierra Information Services
www.**SierraInformation**.com
19229 Sonoma Hwy PMB 264
Sonoma, CA 95476
(707) 996 7380 voice
Email: Andrew@sierrainformation.com

Steps to Success with PROC MEANS

- Very powerful BASE SAS Procedure
- Analyzes *numeric variables*
- Analyses (Output) stored in
  - Output Window (Default)
  - SAS Data Sets (Optional)
- Why Use PROC MEANS?

Why Use PROC MEANS?

- Analyze numeric variables
- Prepare “data marts” or analysis data sets for subsequent analyses
- Generate reports
- Create SAS data sets for use in other tasks
- Explore data prior to applying other SAS capabilities

PROC MEANS: Basic Steps

- PROC MEANS is a BASE SAS Procedure
- Identical to PROC SUMMARY as of Version 6.0 (released in 1991)
- Many features/enhancements added in SAS 8 and SAS 9 Software
- **Defaults**:
  - Analyzes all numeric variables in a SAS data set
  - Presents results in the Output Window

PROC MEANS: Core Concepts

- **Input Data Set**: SAS data set PROC MEANS will analyze
- **Analysis Variables**: Numeric Variables whose values will be analyzed by PROC MEANS
- **Statistics Keywords**: The statistical analyses PROC MEANS will generate
- **Output Data Set**: SAS data set created by PROC MEANS containing the analyses (optional)
- **Classification Variables**: Numeric or character variables whose values will be used to “classify” or “subgroup” the analyses

Step 1: Getting Started

- Example Data Set:
  - Electrical Utility Customer Data Set
  - About 16,000 Observations
    - One Year (12 Months) of Data About Customer Use of/Payment for Electricity
    - KwH: Kilowatt Hour (basic unit of electric consumption)
    - Revenue: Amount charged customer for KwH consumed
    - 12 Monthly Variables

Step 1: Getting Started

- Task: Analyze January KwH Usage

```
* step 1:  the basics;
* default analyses of KwH1;
* KwH1 = January Consumption;
options nonumber nocenter nodate;
title 'SUGI 29 Tutorials Section';
title2 'Steps for Success with PROC MEANS';

proc means data=sugi29.electricity;
var kwh1;
title3 'Default Analyses and Output';
run;
```

Input Data Set

Analysis Variable

Step 1: Getting Started

```
SUGI 29 Tutorials Section
Steps for Success with PROC MEANS
Default Analyses and Output

The MEANS Procedure

                Analysis Variable : KWH1

     N            Mean         Std Dev         Minimum         Maximum
-------------------------------------------------------------------------
 16238     538.4494396         1036.51               0        65557.00
-------------------------------------------------------------------------
```

Five Default Statistics: N, Mean, Std Dev, Minimum, Maximum

Step 1: Getting Started

- What Happened:
  - PROC MEANS performed the default analyses of KwH1 and placed them in the Output Window
  - All non-missing values of KwH1 in the input data set were used to calculate the analyses

Step 2: Take Control!

- PROC MEANS has many features/options you can use to control
  - Which Observations Are Analyzed
  - What Analyses are Performed
  - Where the Analyses are Presented/Stored

Step 2: Take Control!

- **Task**: Analyze January KwH Usage
  - Just for Customers in the Eastern Region
  - Desired Analyses:
    - MEAN, MEDIAN, N, NMISS
- **Solutions**:
  - WHERE Clause Data Set Option
  - Statistics Keywords

Step 2: Take Control!

```
proc means
 mean median n nmiss
      data=sugi29.electricity
      (where= (region = 'EASTERN') );
var KwH1;
label kwh1 = 'January KwH Usage';
title3 'Selecting Observations to Analyze';
title4 'and Statistical Analyses';
title5 'Eastern Region';
run;
```

Statistics Keywords

WHERE Clause Data Set Option

Step 2: Take Control!

```
SUGI 29 Tutorials Section
Steps for Success with PROC MEANS
Selecting Observations to Analyze
and Statistical Analyses
Eastern Region

The MEANS Procedure

  Analysis Variable : KWH1 January KwH Usage

                                          N
       Mean          Median          N  Miss
-------------------------------------------------
 489.0035454   350.0000000       5077    47
-------------------------------------------------
```


Step 2: Take Control!

- What Happened?
  - PROC MEANS calculated the requested statistical analyses
  - The WHERE Clause SAS Data Set Option was used to select observations from the Input Data Set to Analyze
  - Results placed in Output Window

Descriptive Statistics Keywords

This table shows the descriptive statistics available from PROC MEANS starting in Version 8 of the SAS System. Quantile and hypothesis testing keywords are shown on the next two slides

CLM	RANGE
CSS	SKEWNESS |SKEW
CV	STDDEV| STD
KURTOSIS| KURT	STDERR
LCLM	SUM
MAX	SUMWGT
MEAN	UCLM
MIN	USS
N	VAR
NMISS	

Quantile Statistics Keywords

Quantile statistics were added to PROC MEANS in Version 8. A common set of analysis keywords is now available in PROCs MEANS, SUMMARY, REPORT, TABULATE and UNIVARIATE

MEDIAN| P50	Q3| P75
P1	P90
P5	P95
P10	P99
Q1|P25	QRANGE

Hypothesis Testing Keywords

PROBT	T

There are a total of 31 Keywords avaialble in PROC MEANS, each of which can provide an analysis of numeric variables in your data sets.

Step 3: Take Even More Control!

- Tasks:
  - Analyze the values of six analysis variables
    - SUM, MEAN and NMISS
  - Round the analyses to two decimal places
  - Show the variable labels in the PROC MEANS-generated output

Step 3: Take Even More Control!

```
* step 3: take even more control!;
proc means data=sugi29.electricity
           sum n nmiss maxdec=2;
var kwh1-kwh3 rev1-rev3;
label kwh1 = 'Jan KwH'
      kwh2 = 'Feb KwH'
      kwh3 = 'Mar KwH'
      rev1 = 'Jan Rev'
      rev2 = 'Feb Rev'
      rev3 = 'Mar Rev';
title3 'Take Even More Control!';
run;
```

Statistics Keywords

MAXDEC Option

LABEL Statement

Step 3: Take Even More Control!

```
SUGI 29 Tutorials Section
Steps for Success with PROC MEANS
Take Even More Control!

The MEANS Procedure
```

Variable	Label	Sum	N	N Miss
KWH1	Jan KwH	8743342.00	16238	143
KWH2	Feb KwH	7841246.00	16275	106
KWH3	Mar KwH	7671366.00	16265	116
REV1	Jan Rev	1035063.11	16243	138
REV2	Feb Rev	935605.48	16275	106
REV3	Mar Rev	909575.34	16278	103

Step 4: Use a CLASS Variable

- Requesting Sub-Group Analyses Using the **CLASS** or **BY** Statement
  - **CLASS**: Input data set does not have to be sorted by the values of the Classification Variable (or Variables)
  - **BY**: Input Data set is already sorted by the values of the variables used to generated sub-group analyses

Step 4: Use a CLASS Variable

- CLASS versus BY Statement
  - Use the BY Statement when
    - Input Data Set is already sorted by the values of your Classification Variables
    - You Have a "Large" Data Set
      - Sort the Data Set First
      - BY is more efficient than CLASS with "large" Data Sets

Step 4: Use a CLASS Variable

```
* step four:  Be CLASS(y)!;
proc means data=sugi29.electricity
                   mean median
                   maxdec=0;
class region ;
var rev6 kwh6;
title4 'Be CLASS(y)!';
run;
```

Classification Variable

Step 4: Use a CLASS Variable

```
SUGI 29 Tutorials Section
Steps for Success with PROC MEANS
Take Even More Control!
Be CLASS(y)!

The MEANS Procedure
```

REGION	N Obs	Variable	Mean	Median
EASTERN	5124	REV6	69	49
		KWH6	562	411
NORTHERN	5462	REV6	76	60
		KWH6	625	501
SOUTHERN	720	REV6	60	53
		KWH6	508	448
WESTERN	5075	REV6	66	54
		KWH6	531	447

Step 4: Use a CLASS Variable

- Results:
  - Analysis by values of the Classification Variable, REGION
  - Results in ascending order (lowest to highest value) of Classification Variable
  - Column called “N Obs” (number of observations) shown by default
    - Use the NONOBS Option in the PROC MEANS Statement if you don’t want this column in your Output

25

Step 5: Use Multiple Class Variables and Order Output

- Tasks:
  - Obtain MAX, MIN and MEAN of REV7 By:
    - REGION *and* TRANS (Transformer Model)
      - For Transformers in the “K” Series
  - Order Output Results By
    - Frequency of TRANS within REGION

26

Step 5: Use Multiple Class Variables and Order Output

```
* step five: use multiple class variables;
proc means
data=sugi29.electricity(where=
     (cesched IN('E1','E1L','E1M')
     and trans =: 'K'))
     max min mean maxdec=2 ;
class region;
class trans;
var KWH7;
title4 'Using Two Class Variables';
title5 'Without Order=FREQ';
run;
```

27

Step 5: Use Multiple Class Variables and Order Output

SUGI 29 Tutorials Section
Steps for Success with PROC MEANS

Using Two Class Variables
Without Order=FREQ

The MEANS Procedure

Analysis Variable : KWH7

REGION	TRANS	N Obs	Maximum	Minimum	Mean
EASTERN	K1211C	84	1589.00	30.00	470.68
	K1233C	409	2493.00	0.00	483.75
NORTHERN	K1211C	148	3443.00	112.00	627.72
	K1233	5	595.00	595.00	595.00
	K1233C	439	1368.00	9.00	524.29
SOUTHERN	K1211C	18	905.00	321.00	564.17
	K1233C	62	1236.00	0.00	468.42
WESTERN	K12	4	1282.00	1282.00	1282.00
	K1211C	116	915.00	18.00	421.09
	K1233C	377	1361.00	9.00	488.05

Results by Ascending Value of TRANS *within* Ascending Value of Region

28

Step 5: Use Multiple Class Variables and Order Output

- Ordering Options in the CLASS Statement
  - New to Version 8
  - Default is Order=INTERNAL
    - Ascending order of the values of analysis variables as they are stored “internally” (that is, in the data set
  - Options: DESCENDING, ORDER=FREQ, ORDER=FORMATTED

29

Step 5: Use Multiple Class Variables and Order Output

```
* step five: use multiple class variables;
proc means
data=sugi29.electricity(where=
     (cesched IN('E1','E1L','E1M')
     and trans =: 'K'))
     max min mean maxdec=2 ;
class region;
class trans/ORDER=FREQ;
var KWH7;
title4 'Using Two Class Variables';
title5 'With Order=FREQ';
run;
```

30

Step 5: Use Multiple Class Variables and Order Output

SUGI 29 Tutorials Section
Steps for Success with PROC MEANS

Using Two Class Variables
With Order=FREQ

The MEANS Procedure

Results by *Descending Frequency* of TRANS *within* Ascending Value of Region

Analysis Variable : KWH7

REGION	TRANS	N Obs	Maximum	Minimum	Mean
EASTERN	K1233C	409	2493.00	0.00	483.75
	K1211C	84	1589.00	30.00	470.68
NORTHERN	K1233C	439	1368.00	9.00	524.29
	K1211C	148	3443.00	112.00	627.72
	K1233	5	595.00	595.00	595.00
SOUTHERN	K1233C	62	1236.00	0.00	468.42
	K1211C	18	905.00	321.00	564.17
WESTERN	K1233C	377	1361.00	9.00	488.05
	K1211C	116	915.00	18.00	421.09
	K12	4	1282.00	1282.00	1282.00

Step 6: Don't miss the MISSINGS!

- Task:
  - Obtain Count of KwH12 for Customers Whose Electricity is Delivered Via the "K" Series of Transformers
  - Include Observations with missing values for KwH12 in the output analysis.

Step 6: Don't miss the MISSINGS!

```
proc format;
   value group1f
   0 - 500 = '1) 0 to 500 KwH'
 501 - 1000 = '2) 501 to 1,000 KwH'
1001 - 2000 = '3) 1,001 to 2,000 KwH'
2001 - high = '4) 2,001 KwH and Above';
      run;
```


Step 6: Don't miss the MISSINGS!

```
* Step 6: Don't Miss the Missings!;
proc means
data=sugi29.electricity(where=
     (trans=: 'K'))
     maxdec=0
     n mean sum nonobs;
class trans;
class kwh12;
var kwh12;
format kwh12 group1f.;
title4 "Don't Miss the Missings!";
run;
```

KWH12 is both a Classification <u>and</u> an Analysis Variable in this PROC MEANS task. The values of KWH12 will be analyzed <u>within</u> the *<u>formatted</u>* <u>classes</u> of KWH12.

Step 6: Don't miss the MISSINGS!

By default, observations with missing values of the Classification Variable(s) are not used by PROC MEANS to create the output analysis.

SUGI 29 Tutorials Section
Steps for Success with PROC MEANS

Don't Miss the Missings!

The MEANS Procedure

Analysis Variable : KWH12

TRANS	KWH12	N	Mean	Sum
K1211C	1) 0 to 500 KwH	248	283	70206
	2) 501 to 1,000 KwH	80	690	55164
	3) 1,001 to 2,000 KwH	38	1254	47665
	4) 2,001 KwH and Above	8	3793	30341
K1233C	1) 0 to 500 KwH	765	291	222597
	2) 501 to 1,000 KwH	424	670	284022
	3) 1,001 to 2,000 KwH	112	1365	152838
	4) 2,001 KwH and Above	7	2260	15818

Step 6: Don't miss the MISSINGS!

```
* Step 6: Don't Miss the Missings!;
proc means
data=sugi29.electricity(where=
     (trans=: 'K'))
     maxdec=0 MISSING
     n mean sum nonobs;
class trans;
class kwh12;
var kwh12;
format kwh12 group1f.;
title4 "Don't Miss the Missings!";
run;
```

Step 6: Don't miss the MISSINGS!

In this example, using the MISSING Option identified two K-Series transformer models through which no power flowed in December.

```
SUGI 29 Tutorials Section
Steps for Success with PROC MEANS
Don't Miss the Missings!
The MEANS Procedure

                    Analysis Variable : KWH12

TRANS      KWH12                          N          Mean           Sum
-------------------------------------------------------------------------
K12                                       .             0             .             .
K1211C                                    .             0             .             .
-------------------------------------------------------------------------
           1) 0 to 500 KwH              248           283         70206
           2) 501 to 1,000 KwH           80           690         55164
           3) 1,001 to 2,000 KwH         38          1254         47665
           4) 2,001 KwH and Above         8          3793         30344
-------------------------------------------------------------------------
K1233                                     .             0             .             .
K1233C                                    .             0             .             .
-------------------------------------------------------------------------
           1) 0 to 500 KwH              765           291        222597
           2) 501 to 1,000 KwH          424           670        284022
           3) 1,001 to 2,000 KwH        112          1365        152838
           4) 2,001 KwH and Above         7          2260         15818
-------------------------------------------------------------------------
```


Recap I

- So far, we have looked at PROC MEANS capabilities to
  - Conduct analyses we select using Statistics Keywords
  - Sub-group, or classify the results using Classification Variables
    - Order the output analyses in the Output Wndow
  - Round the output analyses to the desired number of decimal places
  - Include observations with missing values in the output analysis

Creating Output SAS Data Sets with PROC MEANS

- PROC MEANS can store your analyses in output SAS data sets. This is useful when you:
  - Want to export the analyses to another file format (e.g., to an Excel™ spreadsheet)
  - Create "lightly summarized" data sets/data marts for use at later points in your project/analysis
  - Use the output in subsequent reporting/analysis projects

Using PROC MEANS to Create Output SAS Data Sets

- Initial Considerations:
  - The NOPRINT Option in the PROC MEANS Statement suppresses display of analyses in the Output Window
  - The OUTPUT Statement controls the creation of a SAS data set by PROC MEANS
  - First, the default data set

Step 7: The Default Output SAS Data Set

```
* Step 7: The Default Output Data Set;
proc means data=sugi29.electricity
                                    NOPRINT;
var kwh12;
output out = out1;
run;

proc print data=out1;
title3 'Data Set Out1';
run;
```


Step 7: The Default Output SAS Data Set

```
SUGI 29 Tutorials Section
Steps for Success with PROC MEANS
Data Set Out1

Obs    _TYPE_    _FREQ_    _STAT_        KWH12

 1        0       16381     N             16002
 2        0       16381     MIN               0
 3        0       16381     MAX           29280
 4        0       16381     MEAN     529.80458693
 5        0       16381     STD      772.59267747
```

Step 8: Select Your Analyses

- Building Your Own Output Data Set
  - Use Statistics Keywords and Other Options in the Output Statement to:
    - **Select** which analyses you want performed on which analysis variables
    - **Assign** names to the variables in the output data set which contain the requested analyses
    - **Use** Data Set Options to further customize/control the data set creation process

Step 8: Select Your Analyses

```
* step 8: select your own analyses;
proc means data=sugi29.electricity NOPRINT;
class region;
var kwh12;
output out=out2 sum  = sumkwh12
                mean = meankwh12;
run;
```


Step 8: Select Your Analyses

Variables "holding" the analyses

REGION	_TYPE_	_FREQ_	sumkwh12	meankwh12
	0	16381	8477933	529.80458693
EASTERN	1	5124	2498596	500.01921153
NORTHERN	1	5462	3256173	609.31380988
SOUTHERN	1	720	510950	721.68079096
WESTERN	1	5075	2212214	446.64122754

Step 8: Select Your Analyses

- Automatic Variables
- **_FREQ_ :** number of observations in the Input Data Set whose values were used to calculate the analyses in that row/observation of the Output Data Set
- **_TYPE_:** information about the Classification Variables used to form that observation in the Output Data Set

Step 9: Be Even More Selective

- You can request different analyses of the analysis variables.
- For example, obtain the
  - Mean and Median of REV6
  - Max and Min of REV7
  - Classified by REGION and "K-Series" transformers

Step 9: Be Even More Selective

```
* step 9: be more selective;
proc means noprint data=sugi29.electricity(where=
                        (region
IN('EASTERN','WESTERN') and trans=: 'K'));
class region trans;
var rev6 rev7;
output out=out3   mean(rev6) = meanrev6
                  median(rev6)=medrev6
                  max(rev7) = maxrev7
                  min(rev7) = minrev7 ;
run;
```

Step 9: Be Even More Selective

REGION	TRANS	_TYPE_	_FREQ_	meanrev6	medrev6	maxrev7	minrev7
		0	1013	63.190889615	57.15	339.01	5
	K12	1	4	195.13	195.13	172.16	172.16
	K1211C	1	201	58.687820896	47.29	206.38	5
	K1233C	1	808	63.657919777	58.37	339.01	5
EASTERN		2	512	63.755572852	56.975	339.01	5
WESTERN		2	501	62.613808144	57.94	173.03	5
EASTERN	K1211C	3	85	61.943082353	43.67	206.38	5
EASTERN	K1233C	3	427	64.116373068	58.7807	339.01	5
WESTERN	K12	3	4	195.13	195.13	172.16	172.16
WESTERN	K1211C	3	116	56.3025	54.69	121.11	5
WESTERN	K1233C	3	381	63.144115171	58.37	173.03	5

Step 9: Be Even More Selective

- The previous slide shows the Output Data Set created by PROC MEANS
  - The values of Analysis Variables were analyzed at *all possible combinations* of the two Classification Variables
    - By Region (_TYPE_ = 2)
    - By Transformer (_TYPE_ = 1)
    - By Region *and* Transformer (_TYPE_ = 3)
    - Overall (_TYPE_ = 0)

Step 10: Keep What You Need

- If you don't need all of the automatically-generated observations, at all possible values of _TYPE_ in your output data set, use the NWAY Option in the PROC MEANS Statement
  - Only observations with the highest value of _TYPE_ are then placed in the Output Data Set.

Step 10: Keep What You Need

- Task:
  - Perform the previous analysis, but create an Output SAS Data Set with analyses at the combination of REGION and TRANS

Step 10: Keep What You Need

```
proc means noprint
data=sugi29.electricity(where=
     (region IN('EASTERN','WESTERN')
           and trans=: 'K')) NWAY;
class region trans;
var rev6 kwh7;
output out=out4   mean(rev6) = meanrev6
                  median(rev6)=medrev6
                  max(rev7) = maxrev7
                  min(rev7) = minrev7;
run;
```


Step 10: Keep What You Need

REGION	TRANS	_TYPE_	_FREQ_	meanrev6	medrev6	maxrev7	minrev7
EASTERN	K1211C	3	85	61.943082353	43.67	206.38	5
EASTERN	K1233C	3	427	64.116373068	58.7807	339.01	5
WESTERN	K12	3	4	195.13	195.13	172.16	172.16
WESTERN	K1211C	3	116	56.3025	54.69	121.11	5
WESTERN	K1233C	3	381	63.144115171	58.37	173.03	5

Step 11: Create Multiple SAS Data Sets in One PROC MEANS Task

- You can place multiple Output Statements in one PROC MEANS Task
  - Much more efficient than "interrogating" the Input Data Set multiple times.
  - Instead, read it once, and create all the data sets you need in a single PROC MEANS step.

55

Step 11: Create Multiple SAS Data Sets in One PROC MEANS Task

- Example:
  - Five Classification Variables
    - REGION, OFFICE, TRANS, CESCHED, SERIAL
    - 32 values of _TYPE_ (0 to 31)
  - By testing the value(s) of _TYPE_ in a WHERE Clause, we can direct the placement of analyses conducted by PROC MEANS in to separate data sets

56

Step 11: Create Multiple SAS Data Sets in a Single PROC MEANS Task

- Understanding the CHARTYPE Option
  - New to Version 8
  - Converts the default numeric value(s) of _TYPE_ to a character string representing the binary value of the variable
    - Length is equal to the number of classification variables
  - Simplifies creation of multiple Output Data Sets in a single PROC MEANS task.

57

Step 11: Create Multiple SAS Data Sets in a Single PROC MEANS Task

```
proc means noprint chartype descendtypes
      data=sugi29.electricity ;
class region office trans cesched serial;
var kwh10 rev10;
output out=out5(where=(_type_ in('00000','10000'))) sum=/autoname;
output out=out6(where=(_type_ in('00000','11010'))) sum=/autoname;
output out=out7(where=(_type_ in('00011'))) sum=/autoname;
output out=out8(where=(_type_ in('00100','00000'))) sum=autoname;
run;
```

58

Step 11: Create Multiple SAS Data Sets in a Single PROC MEANS Task

```
835  proc means noprint chartype descendtypes
836       data=sugi29.electricity ;
837  class region office trans cesched serial;
838  var kwh10 rev10;
839  output out=out5(where=(_type_ in('00000','10000'))) sum=/autoname;
840  output out=out6(where=(_type_ in('00000','11010'))) sum=/autoname;
841  output out=out7(where=(_type_ in('00011'))) sum=/autoname;
842  output out=out8(where=(_type_ in('00100','00000'))) sum=autoname;
843  run;

NOTE: There were 16381 observations read from the data set
SUGI29.ELECTRICITY.
NOTE: The data set WORK.OUT5 has 5 observations and 9 variables.
NOTE: The data set WORK.OUT6 has 123 observations and 9 variables.
NOTE: The data set WORK.OUT7 has 111 observations and 9 variables.
NOTE: The data set WORK.OUT8 has 21 observations and 8 variables.
```

59

Step 11: Create Multiple SAS Data Sets in a Single PROC MEANS Task

Using the CHARTYPE Option converted the Default numeric values of _TYPE_ into a

TRANS	_TYPE_	_FREQ_	KWH10_Sum	REV10_Sum
K12	00100	2	2592	346.8
K1211C	00100	1	170490	21713.2813
K1233	00100	5	2440	307.5
K1233C	00100	1326	593782	73998.91965
	00000	1717	769585	96858.50095

Places observations with different combinations of the Classification Variables in to different Output Data Sets.

The DESCENDTYPES Option places observations in Output Data Sets in descending value of _TYPE_. The effect is to put the "Type Zero" observation at the bottom of the Output SAS Data Set.

60

Step 11: Create Multiple SAS Data Sets in a Single PROC MEANS Task

TRANS	_TYPE_	_FREQ_	KWH10_Sum	REV10_Sum
K12	00100	4	2592	346.8
K1211C	00100	382	170490	21713.2813
K1233	00100	5	2440	307.5
K1233C	00100	1326	593782	73998.91965
	00000	1717	769304	96366.50095

Additional PROC MEANS Capabilities

- PRELOADFMT and COMPLETETYPES Options
- CLASSDATA Data Sets
- IDGROUP Option
- TYPES Statement
- WAYS Statement
- AUTOLABEL Option
- Use of Multilabel Formats
- Threading (SAS 9)

Thanks for Attending This Presentation!

- Copies of this and other SAS user group presentations available at www.SierraInformation.com
  - Click on the “past presentations” link on the upper left-hand corner of the homepage.

SIERRA INFORMATION SERVICES

...Maximizing your investment in the SAS® System... and those in your organization who use it...